
1 Abstract

Methods of generating multivariate normal data are discussed, statistical and graphic
methods are used to check the generated data sets. Codes of both non-macro version
and macro version are provided, some SAS ”tricks” are provided as well.

2 Introduction

In simulation studies in statistics, there are many situations that we need to generate
data from a multivariate normal distribution. By multivariate normal data we mean
joint observations of p variables Y1, Y2, . . . , Yp, in which each individual variable by itself
is normally distributed, the variables are mutually correlated, and come from a joint
multivariate normal distribution. The procedure for generation of multivariate normal
data is similar to the univariate case, that is, we can generate pairs of independent normals
and then multiplied that pairs by the Cholesky square root of the desired variance-
covariance matrix. One way to do that is to obtain the formula for the Cholesky square
root of the variance-covariance matrix, and which is easy for bivariate normal data.
However, it becomes complicated when p is large. An alternative method by using PROC
IML can be used to accomplish the desired data easily by using matrix computations.
Function HALF in PROC IML can be used to obtain the Cholesky square root of the
desired variance-covariance matrix.

3 Data generation and the corresponding codes

3.1 Generate the bivariate normal data

As mentioned in introduction, obtaining the formula for the Cholesky square root of the
desired variance-covariance matrix is easy for bivariate normal data, we introduced it as
a special case for the multivariate normal data when p = 2.

Suppose that we want to obtain that
(

Y1

Y2

)
∼ N

((
µ1

µ2

)
, Σ

)
,

where the variance-covariance matrix Σ is

Σ =

(
σ2

1 ρσ1σ2

ρσ1σ2 σ2
2

)
,

then we can obtain Σ1/2 as

Σ1/2 =

(
σ1 0

ρσ2

√
σ2

2(1− ρ2)

)
,

so Y1 = µ1 + σ1 ∗ rannor1 and Y2 = µ2 + ρ ∗ σ2 ∗ rannor1 +
√

σ2
2(1− ρ2) ∗ rannor2,

where rannor1 and rannor2 are two independent random variables. Thus, we can use
the following code to generate the bivariate normal data:

/* Generate the bivariate normal data */

data one;

mean1=0; *mean for y1;

mean2=10; *mean for y2;

sig1=2; *SD for y1;

sig2=5; *SD for y2;

rho=0.5; *Correlation between y1 and y2;

do i = 1 to 1000;

r1 = rannor(1245);

r2 = rannor(2923);

y1 = mean1 + sig1*r1;

y2 = mean2 + rho*sig2*r1+sqrt(sig2**2-sig2**2*rho**2)*r2;

output;

end;

keep y1 y2;

*proc print;

run;

Such that Y1 and Y2 are bivariate normally distributed.

3.2 Generate the multivariate normal data by using PROC IML

As discussed, obtaining the formula for the Cholesky square root of the desired variance-
covariance matrix is complicated when p is large. Next, we will introduce an alternative
method by using PROC IML.

In matrix notation the random variable is expressed as a vector Y′ = (Y1, Y2, . . . , Yp)
which has a multivariate normal distribution with mean vector µ′ = (µ1, µ2, . . . , µp) and
variance-covariance matrix

Σ =




σ2
1 σ12 . . . σ1p

σ12 σ2
2 . . . σ2p

...
...

. . .
...

σp1 σp2 . . . σ2
p


 ,

each Yi(i = 1, 2, . . . , p) has a N(µi, σ
2
i ) distribution, the covariance of Yi and Yj is σij,

and the correlation between Yi and Yj is ρij =
σij

σiσj
. Now suppose Zi(i = 1, 2, . . . , p) are

independent and have a N(0, 1) distribution. Then the vector Z′ = (Z1, Z2, . . . , Zp) has
mean vector 0′ = (0, 0, . . . , 0), and the covariance between Zi and Zj is 0. Therefore the
variance-covariance matrix for Z′ is the identity matrix.

If A is a symmetric p × p matrix, then the statement T = HALF (A) returns T as
an upper-triangular p× p matrix such that T ′T = A, where A must be positive-definite,
otherwise the HALF function will cause an error which stops the program. This function
is now applied to the variance-covariance matrix of Y.

If T = HALF (Σ), then the transformation Y = µ+T ′Z results in a random vector Y
with the desired multivariate normal distribution. Because each Y is a linear combination
of normal variates, Y is normally distributed. To check that the mean and variance-
covariance are correct, the expected value of Y is E(Y ) = E(µ + T ′Z) = µ + 0 = µ, and
the variance of Y is

V ar(Y ) = E[(Y − E(Y ))(Y − E(Y ))′]
= E[(T ′Z)(T ′Z)′]
= E[T ′ZZ ′T ]
= T ′E[ZZ ′]T
= T ′IT
= T ′T
= Σ

If instead, what we have is a correlation matrix R, the variance-covariance matrix
Σ can be obtained from R by forming a diagonal matrix, D, whose elements are the
standard deviation of each Y . The function to do this in PROC IML is DIAG. Then
we pre-multiply and post-multiply R by D to obtain Σ:

DRD′ =




σ1 0 . . . 0

0 σ2
. . .

...
...

. . . . . . 0
0 . . . 0 σp


×




1 ρ12 . . . ρ1p

ρ12 1
. . .

...
...

. . . . . . ρp−1,p

ρ1p . . . ρp−1,p 1


×




σ1 0 . . . 0

0 σ2
. . .

...
...

. . . . . . 0
0 . . . 0 σp


 = Σ.

3.2.1 Code of generating the multivariate normal data by using PROC IML
- nonmacro version

The program in this subsection is a non-macro version program which generates 1000
observations of the variables Y1, Y2, Y3, beginning with the correlation matrix R and a
vector of means µ = (µ1, µ2, µ3)

′ and standard deviations σ = (σ1, σ2, σ3)
′ read instream

as variables using a CARDS statement.

/* Generate the multivariate normal data in SAS/IML */

/* non-macro version */

data MVN_par; /* data for the parameter for the multivariate normal data */

input r1 r2 r3 means vars;

cards;

1.0 -0.5 0.9 100 2

-0.5 1.0 -0.7 200 5

0.9 -0.7 1.0 300 10

;

proc iml;

use MVN_par;

read all var {r1 r2 r3} into R;

read all var {means} into mu;

read all var {vars} into sigma;

p = ncol(R);

diag_sig = diag( sigma );

DRD = diag_sig * R * diag_sig‘;

U = half(DRD);

do i = 1 to 1000;

z = rannor( j(p,1,1234));

y = mu + U‘ * z;

yprime = y‘;

yall = yall // yprime;

end;

varnames = { y1 y2 y3 };

create my_MVN from yall (|colname = varnames|);

append from yall;

quit;

proc print data=my_MVN;

run;

3.2.2 Code of generating the multivariate normal data by using PROC IML
- macro version

The program in this subsection is a macro version program which is good for any di-
mension multivariate normal data with variance-covariance matrix and means as macro
arguments.

%macro mvn(varcov=, means=, n=, myMVN=);

/* arguments for the macro:

1. covcov: data set for variance-covariance matrix

2. means: data set for mean vector

3. n: sample size

4. myMVN: output data set name */

proc iml;

use &varcov; /* read in data for variance-covariance matrix */

read all into sigma;

use &means; /* read in data for means */

read all into mu;

p = nrow(sigma); /* calculate number of variables */

n = &n;

l = t(half(sigma)); /* calculate cholesky root of cov matrix */

z = normal(j(p,&n,1234)); /* generate nvars*samplesize normals */

y = l*z; /* premultiply by cholesky root */

yall = t(repeat(mu,1,&n)+y); /* add in the means */

varnames = { y1 y2 y3 };

create &myMVN from yall (|colname = varnames|);

append from yall;

quit;

%mend mvn;

data means1;

input x @@;

cards;

100 200 300

;

run;

data varcov1;

input x1-x3;

cards;

4 -5 18

-5 25 -35

18 -35 100

;

run;

%mvn(varcov=varcov1, means=means1, n=1000, myMVN=my_MNV)

proc print data=my_MNV;

run;

3.3 Statistical and graphic methods to check the generated mul-
tivariate normal data

Several procedures can be used to illustrate that the data sets generated have the desired
properties, including statistical and graphic methods. For example, means, standard
deviation and normality test for each variable (PROC UNIVARIATE), pairwise corre-
lations (PROC CORR), bar charts of each variable (PROC GCHART), and pairwise
plots (PROC GPLOT). Follows are the codes to check the generated data from previous
section.

proc univariate normal noprint data=test1;

var y1 y2 y3;

output out=new mean=avg1 avg2 avg3 std=std1 std2 std3 probn=prob1 prob2 prob3;

run;

proc print data=new;

run;

proc corr data=test1 noprint outp=mycorr;

var y1 y2 y3;

run;

proc print data=mycorr;

run;

goptions reset=all;

symbol1 value=dot cv=blue height=0.5 width=2;

proc gplot data=test1;

plot y1*y2 y2*y3 y1*y3;

run;

proc gchart data=test1;

vbar y1 y2 y3;

run;

4 Conclusion

This paper discussed the methods of generation of data from multivariate normal dis-
tribution, the function HALF in PROC IML can be easily used to obtain the Cholesky
square root of the variance-covariance matrix, and then obtain the desired multivariate
normal data. The codes are useful for any dimensional multivariate normal data, as long
as you have the parameters for the distribution, you can use the macro to obtain the
data you want. If the sample size for the data set are large, use ”Batch Submit with
SAS” can be faster and obtain the results sooner.

